Question: I am preparing a figure for an article, and I would like very much to be able to fill a curve, but not under it, but to the vertical axis, i.e. to the left. Due to the meaning of the data I present, I cannot exchange the axes.

Here is a minimal example:

```
Plot[ArcCot[x], {x, 0, 1}, PlotRange -> {{0, 1.1}, {0.5, 2}}, AxesLabel -> {x, y}]

```

Can anyone think of a way of filling the curve to the`y` axis? I find no mention of this kind of filling in the standard documentation.

Answer: 
Adding a line at the minimum value and shading as normal:-

```
min = MinValue[ArcCot[x], 0 <= x <= 1, x];
Plot[{ArcCot[x], min}, {x, 0, 1}, PlotRange -> {{0, 1.1}, {0.5, 2}}, 
 AxesLabel -> {x, y}, Filling -> {1 -> {2}}, PlotStyle -> {Automatic, None}]

```

![enter image description here](https://i.stack.imgur.com/2EIdA.png)
If you mean to fill everything to the right of the curve (you wrote "to the right", did not you?), try this:

```
    RegionPlot[y > ArcCot[x], {x, 0, 1.1}, {y, 0.5, 2}, 
 AxesLabel -> {x, y}]

```

![enter image description here](https://i.stack.imgur.com/zIqKm.jpg)

In response to your doubts: all this in your hands.
This is without extra axes:

```
  RegionPlot[y > ArcCot[x], {x, 0, 1.1}, {y, 0.5, 2}, AxesLabel -> {x, y}, FrameTicks -> {Automatic, Automatic, None, None}] 

```

![enter image description here](https://i.stack.imgur.com/KL2uV.jpg)

This is without off-set:

```
RegionPlot\[y > ArcCot\[x\], {x, 0, 1.1}, {y, 0.5, 2}, 
 AxesLabel -> {x, y}, PlotRangePadding -> 0\]][3]

```

![enter image description here](https://i.stack.imgur.com/tcAMQ.jpg)
This is with a blue boundary, rather than gray one:

RegionPlot[y > ArcCot[x], {x, 0, 1.0}, {y, 0.5, 2}, 
 AxesLabel -> {x, y}, PlotRangePadding -> 0, 
 FrameTicks -> {Automatic, Automatic, None, None}, 
 Frame -> {True, True, False, False}, 
 BoundaryStyle -> {Blue, Thickness[0.005]}, 
 PlotRange -> {{-0.1, 1.1}, {-0.1, 2.1}}]

![enter image description here](https://i.stack.imgur.com/NTZSz.jpg)

and this is, if you like to have in blue only the line ActCot[x]:

```
    Show[{
  RegionPlot[y > ArcCot[x], {x, 0, 1.0}, {y, 0.5, 2}, 
   AxesLabel -> {x, y}, PlotRangePadding -> 0, 
   FrameTicks -> {Automatic, Automatic, None, None}, 
   Frame -> {True, True, False, False}, 
   PlotRange -> {{-0.1, 1.1}, {-0.1, 2.1}}],
  Plot[ ArcCot[x], {x, 0, 1.0}, PlotStyle -> {Blue, Thickness[0.005]}]
  }]

```

![enter image description here](https://i.stack.imgur.com/BJ4qd.jpg)

Have fun!